# Series expansion of integral

I'm looking for a way to do a series expansion of

$$\frac{\mu_0bI_0}{4\pi}\int_0^{2\pi}\frac{\cos\left[\omega\left(t-\frac{1}{c}\sqrt{r^2+b^2-2rb\sin(\theta)\cos(\phi)}\right)\right]}{\sqrt{r^2+b^2-2rb\sin(\theta)\cos(\phi)}}\cos(\phi)\,d\phi$$in terms of $$b$$ where $$b\ll1$$. I have a solution where I know this simplifies to $$\frac{b^2}{4c}\mu_0I_0\sin(\theta)\frac{1}{r^2}\left[c\cos(\omega t)\cos\left(\frac{1}{c}\omega r\right)+c\sin(\omega t)\sin\left(\frac{1}{c}\omega r\right)-\omega r\sin(\omega t)\cos\left(\frac{1}{c}\omega r\right)+\omega r \cos(\omega t)\sin\left(\frac{1}{c}\omega r\right)\right]+O(b^4)$$but I'm not sure how to manipulate things in Mathematica to do this. I've tried so far to do an expansion of the integrand passing assumptions such as $$r>0$$, $$c>0$$, $$\theta>0$$, $$b>0$$, etc. but nothing seems to really get me to where I need to go.

The code for the integral:

Integrate[
Cos[ω (t -
1/c Sqrt[r^2 + b^2 - 2 r b Sin[θ] Cos[φ]])]/
Sqrt[r^2 + b^2 -
2 r b Sin[θ] Cos[φ]] Cos[φ], {φ, 0,
2 π},
Assumptions -> {r > 0, t > 0, b > 0, ω > 0, θ > 0}]


You could use AsymptoticIntegrate:

AsymptoticIntegrate[
Cos[ω (t-1/c Sqrt[r^2+b^2-2 r b Sin[θ] Cos[φ]])]/Sqrt[r^2+b^2-2 r b Sin[θ] Cos[φ]] Cos[φ],
{φ,0,2 π},
{b,0,3},
Assumptions->r>0
] //FullSimplify


(1/(8 c^3 r^4))b π Sin[θ] (c Cos[(-(r/c) + t) ω] (8 c^2 r^2 + 4 b^2 (-3 c^2 + r^2 ω^2) + 3 b^2 (5 c^2 - 2 r^2 ω^2) Sin[θ]^2) + r ω (4 c^2 (3 b^2 - 2 r^2) + b^2 (-15 c^2 + r^2 ω^2) Sin[θ]^2) Sin[(-(r/c) + t) ω])

• I didn't know about the AsymptoticIntegrate function, thanks for pointing this one out. – Bo Johnson Apr 4 at 13:17

Series-expansion of the integrand: (to any order you like)

A = Assuming[r > 0,
Series[Cos[ω (t - Sqrt[b^2 + r^2 - 2 b r Cos[φ] Sin[θ]]/c)]/Sqrt[b^2 + r^2 - 2 b r Cos[φ] Sin[θ]],
{b, 0, 2}]] //Normal


then integrate:

B = Integrate[A, {φ, 0, 2π}]


$$\frac{\pi(\cos(\omega(t-\frac{r}{c})) (2 b^2 \sin^2(\theta)(3 c^2-r^2 \omega ^2)-4 c^2 (b^2-2 r^2))+b^2 c r \omega(3 \cos(2 \theta)+1) \sin(\omega(t-\frac{r}{c})))}{4 c^2 r^3}$$